
There are several methods available in the fluidization literature for the prediction of the 
porosity of a liquid fluidized bed of monosized spheres. The majority of practical 
applications, however, involve non-spherical particles. Furthermore, adding to the 
complexity of the problem, systems of practical interest usually consist of particles with 
varying sizes. Granular beds used in water and wastewater filtration is an important 
example. My goal in this handout is to explain how the expanded height of a liquid fluidized 
granular bed can be estimated. To this end, the general problem is tackled in three steps, 
looking at three types of problem starting with the simplest and finishing with the most 
complex: (1) Beds of monosized spheres. (2) Beds of monosized non-spherical particles (3) 
Non-uniform beds of non-spherical particles. 

 

MONOSIZED SPHERES 

As noted, there are numerous correlations that can be used for this purpose (Couderc, 1985; 
Hartman et al. , 1989; Di Felice, 1995; Epstein, 2003). Akgiray and Soyer (2006) reviewed and 
compared existing popular correlations and found that an equation they developed yields 
very accurate predictions: 

The dimensionless group φ and the modified Reynolds number Re1 are defined as follows 

(these definitions are valid for both spherical and non-spherical particles; simply set  =1 in 
these expressions to restrict attention to spheres): 

where  

V = velocity based on the empty cross-section of the bed 

 = porosity (i.e. volume fraction of voids) (0 = fixed-bed porosity) 

deq = equivalent diameter defined as the diameter of the sphere with the same volume as a 
bed particle 

 = sphericity defined as the surface area of the equivalent volume sphere divided by the 
actual surface area of the particle 

μ = viscosity of the liquid 

ρ = density of the liquid 



ρp = density of the particles 

g = gravitational acceleration  

 

     

FIGURE 1: Examples of plastic balls (spheres) employed in fluidization experiments carried 

out at the Marmara University Filtration and Fluidization Research Center (FFRC) 

laboratories. (Courtesy of Dr. Elif Soyer.) 

Two types of problem can be considered: (1) Given the value of velocity, what is the 
corresponding porosity? (2) Given the desired porosity, what is the corresponding velocity? 
Usually, instead of the porosity, it is the expanded height L that is of direct practical interest. 

For either type of problem, the fixed-bed porosity 0 and the fixed-bed height L0 are 

normally known in advance. Therefore, the bed height L and the porosity  can be calculated 
from each other using the following relation: 

 L (1-) = L0 (1-0)  (4) 

The best way to understand how these equations are used is to look at a specific example. 

EXAMPLE 1: Glass balls of 4 mm diameter are fluidized using water at a velocity of 0.20 m/s. 
What will be the porosity of the bed? Assume density of glass is 2500 kg/m3, density of 

water is 1000 kg/m3, and viscosity of water is 0.001 N s/m2. Assuming L0 = 1 m and 0 = 0.4, 
what will be the expanded bed height and the percent expansion? 

Solution: 

Using Equations 2 and 3, we write 

Inserting these into Equation 1, we obtain  

Since  appears on both sides of this nonlinear equation, an iterative solution is required. 

Such a solution is carried out easily using the Solver Add-In in MS-Excel. The result is  = 
0.837 (see the screen images given below). Therefore 

  
        

     
 

           

         
       

Percent expansion = 100 (L-L0)/L0 = 268 %. 

Remark: The success of Solver in converging to the correct solution may depend on the 

initial estimate of the expanded bed porosity. In this example, Solver failed when  = 0.60 

was used as the intial value, but it found the correct solution starting from  = 0.75 or  = 
0.90. A solution method that is more robust is presented in Example 3. 

 

 

FIGURE 2-a: An Excel worksheet (formulas made visible by clicking the “Show Formulas” 
button) that can be used for spherical particles (Source: Akgiray and Soyer, 2006). 

 



 

FIGURE 2-b: The use of the Solver Add-In in MS-Excel. The constraints 0.35 ≤  and  ≤ 0.99 
were imposed to assure a physically meaningful solution. The addition of such constraints is 

normally not required, but may be useful if the value of  becomes negative or greater than 
unity during iterations. 

 

MONOSIZED NON-SPHERICAL PARTICLES 

There are very few correlations applicable to non-spherical particles and, as a matter of fact, 
a vast majority of the literature equations are restricted to spheres. Soyer and Akgiray (2009) 
carried out extensive tests with non-spherical filter media and extended Equation 1 to 
handle both spherical and non-spherical particles. The generalized equation is as follows:   

It may be noted that Equation 5 becomes identical with Equation 1 when  = 1. 

EXAMPLE 2: A sieved fraction of silica sand used in fluidization tests had the following 

properties: deq = 1.40 mm, ρp = 2627.6 kg/m3,  = 0.714. Initial bed depth was L0 = 22.9 cm 

and initial porosity was 0 = 0.466 . The bed was fluidized using water at 25.2 oC. Estimate the 
porosity, expanded height and percent expansion of the bed as a function of velocity in the 
range 0.02 m/s to 0.07 m/s. At this temperature, ρ = 997.0 kg/m3, μ = 0.000886 N s/m2. 

Solution: 

First, using Equations 2 and 3, we write 

Inserting these into Equation 5, we obtain  

This equation is to be solved to find   for each given value of V. While such calculations 
could be carried out manually, it is much faster and easier to use a computer for this 
purpose. To solve this problem and other problems similar to it, the worksheet of Figure 2 
was generalized to handle non-spherical media. The resulting worksheet is shown in Figure 
3-a. Solver is executed once for each different value of V. The results are shown in Table 1 
and illustrated in Figure 4. Also shown in Figure 4 are the experimental data for this system 
(Source: Soyer and Akgiray, 2009). 

 

FIGURE 3-a: An Excel worksheet (formulas made visible by clicking the “Show Formulas” 
button) that can be used for both spherical and non-spherical particles. This image was 

captured after the solution (  = 0.469) was found. A reasonable initial value for , e.g. a 

value between 0 and 0.90 should be entered before Solver is executed. 
 



 

FIGURE 3-b: The use of the Solver Add-In in MS-Excel.  

 

Table-1: Calculated results for Example 2. 

velocity (m/sec) porosity L (m) 
Percent 
Expansion 

0.02 0.469 0.230 0.6 

0.03 0.547 0.270 17.9 

0.04 0.613 0.316 38.0 

0.05 0.669 0.369 61.3 

0.06 0.718 0.434 89.4 

0.07 0.761 0.512 123.4 

 

 

 

FIGURE 4: Model predictions versus experimental data for Example 2.  

NON-UNIFORM BEDS OF NON-SPHERICAL PARTICLES  

The majority of real filter media, in addition to being non-spherical in shape, contain a 
distribution of sizes. In practice, a sieve analysis is used to characterize the size distribution 
of a non-uniform granular bed. The “serial model” is currently recommended to predict the 
total expansion of such a bed: A bed with a size gradation is considered to consist of several 
layers of approximately uniform size according to the sieve analysis data, and the expansion 
of each layer is separately calculated. The total expansion is calculated by adding the 
expansions of all the layers (Fair et al., 1971; AWWA, 1999). This calculation method may 
also be used to predict the total expansion of a multimedia bed.  

EXAMPLE 3: (Adopted from Peavy et al., Example 4-14) Water at 20 oC is used to backwash a 
filter bed at a rate of 0.015 m/s. The fixed bed is 0.75 m deep and contains nonuniform sand 
(specific gravity of 2.65). The fixed-bed porosity and sphericity are 0.4 and 0.85, respectively, 
throughout the depth of the bed. Determine the depth of the expanded bed. The size 
distribution of the grains is given below. 

Table-2: Sieve analysis for the filter bed of Example 3. 

Sieve Size (USA mesh) Weight fraction Retaining size (mm) Passing size (mm) 

14 (retaining) 0.01 1.41 
 14-20 0.11 0.84 1.41 

20-25 0.2 0.71 0.84 

25-30 0.32 0.6 0.71 

30-35 0.21 0.5 0.6 

35-40 0.13 0.42 0.5 

40 (passing) 0.02 
 

0.42 

 

Solution: 

This problem can be solved using the worksheet of Example 2 (see Figure 3). Instead of using 
different velocities as in Example 2, however, velocity is kept constant at 0.015 m/s, but 
particle size is varied. A different size (deq) is used for each fraction. This bed is assumed to 
consist of seven uniform layers and the fixed-bed depth of each layer is assumed to be 
proportional to its weight fraction. For the 25x30 mesh fraction, for example, fixed-bed 
depth is calculated as 0.32x0.75 m = 0.24 m (see cell I17 in Figure 7). The fixed-bed depths of 
other fractions are similarly calculated and displayed in Column I (see Figures 7 and 8). 

The topmost sieve retains particles that cannot pass through 1.41 mm openings. Since there 
is no sieve above this sieve, no passing size is specified. The weight fraction (i.e. 0.01) of 
grains retained on this sieve, however, is very small and little error is introduced by assuming 
a passing size equal to the retaining size for this fraction. Similarly, grains passing through 
the 0.42 mm sieve have no retaining size (they accumulate on the tray at the bottom of the 
sieve assembly). Again, using a retaining size equal to the passing size should introduce a 
negligible error in the overall calculation, as the weight fraction (0.02) of the grains falling all 
the way down to the bottom pan (tray) is quite small. 

 



 

FIGURE 5: Sieves placed in the shaking device at  the Marmara University FFRC laboratories. 
There are six sieves and a bottom pan in this picture. The example problem also involves six 
sieves plus a bottom pan. (Courtesy of Ms. Selda Yiğit Hunce.) 

 

 

FIGURE 6: One of the sieves of Figure 5. A crushed glass sample was being sieved manually 
when this picture was taken (Courtesy of Dr. Elif Soyer.) 



 

Equivalent diameters are not reported in this example. (See Soyer and Akgiray (2009) for a 
description of how deq is measured.) A common –but not very good- approximation is to use 
the arithmetic average of the passing and retaining sizes. This approach is adopted in the 
worksheet of Figures 7 and 8 (see the formulas used in Column F). 

At this point, the information in Columns B, C, D, E, F, G and I have been typed in or 
calculated and the next step is the calculation of expanded porosity for each sieve fraction. 
Solver can be executed seven times to find the expanded porosities of the seven fractions 
(the worksheet in Figure 3 can be used for this purpose). The expanded depths of all 
fractions are then calculated and summed to find the total expanded depth of the bed. The 
calculations in Column J are automatic (see the formulas shown in Figure 8) once the 
expanded porosities are calculated. 

A better alternative to the use of Solver is to write a function in VBA to carry out the 
iterations. This can speed up calculations because one would not have to repeat the steps 
needed to execute Solver for each different fraction (or, for each different velocity as in 
Example 2). Copying and pasting the relevant function call (which is written as an Excel 
formula) from one cell to others would be sufficient. Another advantage of this approach is 
that, one can change any of the input parameters (e.g. temperature, particle density, weight 
fractions etc.) and then just press the “Enter” key and the results will be automatically 
calculated. With the first approach, one would have to execute Solver seven additional times 
for each change in any one of the input parameters. The Appendix gives the full listing of a 
VBA code that I wrote just for this purpose. The program in the Appendix seems to work 
correctly and hopefully is easy enough to understand (and to adapt and further develop) by 
others who have some programming skills. Just one warning: This program has not been 
tested much (it was developed within a few hours and tested only with the examples in this 
handout); so if you find what you think is an error or a weakness please let me know. 

 

FIGURE 7: Results for Example 3.  



 

FIGURE 8: The Excel formulas used in the worksheet shown in Figure 7.  



APPENDIX 
 

'Program to calculate fluidized bed expansion 

'Author: Ömer Akgiray, Marmara University 

'Date: May 20, 2014 

 

Option Explicit 

Dim Ga As Double   'Galileo number 

Dim Re As Double   'Reynolds number 

Dim Sph As Double  'Sphericity (same as psi; different name is used here to avoid conflict with psi passed as argument) 

 

Function log10(x As Double) As Double 

    log10 = Application.log10(x) 

End Function 

 

Function logPhi(Re1 As Double, psi As Double) As Double 

    'Correlation by Soyer and Akgiray(2009) 

    'Re1 = modified Reynolds number, psi = sphericity 

    logPhi = log10(3.137 * Re1 + 0.673 * Re1 ^ 1.766) _ 

              - (0.93 + 0.274 * log10(Re1)) * (-log10(psi)) ^ 1.262 

End Function 

 

Function Residual(e As Double) As Double 

    'Function calculates residual error 

    'Root finding routines (e.g. Bisect) try to find e such that Residual = 0 

    'e =  porosity of fluidized bed 

    Dim Re1 As Double, phi As Double 

    Re1 = Re / 6 / (1 - e) 

    phi = e ^ 3 / (1 - e) ^ 2 * Sph ^ 3 * Ga / 216 

    Residual = log10(phi) - logPhi(Re1, Sph) 

End Function 

 

Function por(deq As Double, psi As Double, densp As Double, _ 

             densf As Double, visc As Double, vel As Double) As Double 

    Dim e As Double, Re1 As Double, phi As Double, nb As Integer 

    Dim e1() As Double, e2() As Double, Found As Boolean, Error As Double 

    Const TOL = 0.00000001  'Tolerance for error 

    Re = psi * deq * densf * vel / visc 

    Ga = deq ^ 3 * densf * (densp - densf) * 9.81 / visc ^ 2 

    Sph = psi 

    'We will allow porosities between 0.00001 and 0.99999, but 

    'if the porosity is less than fixed-bed porosity, the bed is 

    'not fluidized and the answer found here should be replaced by 

    'fixed-bed porosity; if porosity is higher than 0.90, then we are 



    'outside the range of applicability of Phi versus Re1 correlations 

    'and the answer may be inaccurate and should be used with caution 

    'Now, bracket one root (we expect only one root) 

    nb = 1 

    zbrak 0.00001, 0.99999, 100, nb, e1, e2, "Residual" 

    If nb = 0 Then  'nb = number of roots found between 0.00001 and 0.99999 

        'nb = 0 means there is no solution for the given input parameters 

        'This may happen if velocity is too high (greater than terminal settling velocity) 

        por = 0 

        Exit Function 

    End If 

    'Find the root between e1(1) and e2(1) 

    Bisect e1(1), e2(1), e, Error, TOL, Found, "Residual" 

    If Found Then 

        por = e 

    Else 

        por = 0 

    End If 

End Function 

 

Sub Bisect(ByVal x1 As Double, ByVal x2 As Double, _ 

           ByRef XMid As Double, ByRef FMid As Double, _ 

           ByVal TOL As Double, ByRef Found As Boolean, _ 

           ByVal Func As String) 

    Dim Itr As Integer 

    Const MAXItr = 50 

    Dim A As Double, B As Double, FA As Double, FB As Double 

    Found = True 

    A = x1 

    B = x2 

    FA = Application.Run(Func, A) 

    FB = Application.Run(Func, B) 

    If FA * FB > 0 Then 

        Found = False 

        Exit Sub 

    End If 

    For Itr = 1 To MAXItr 

      XMid = (A + B) / 2 

      FMid = Application.Run(Func, XMid) 

      If FA * FMid <= 0# Then 

        B = XMid 

        FB = FMid 

      Else 

        A = XMid 



        FA = FMid 

      End If 

      If Abs(B - A) < TOL Then 

        Exit Sub 

      End If 

    Next Itr 

    Found = False 

End Sub 

 

'The following routine finds brackets containing zero crossings 

'Each such bracket can be used as input to Subroutine Bisect 

'See Numerical Recipes in Fortran 77, 2nd Ed., pp.343-345. 

 

Sub zbrak(ByVal x1 As Double, ByVal x2 As Double, _ 

          ByVal n As Integer, ByRef nb As Integer, _ 

          ByRef xb1() As Double, ByRef xb2() As Double, _ 

          ByVal Func As String) 

    'x1, x2: interval end points 

    'n : the interval [x1,x2] is divided into n equal segments 

    'nb: maximum number of roots sought within [x1,x2] 

    '    on return, nb is the number of roots (zero crossings) found 

    'xb1, xb2: bracketing pairs found 

    Dim nbb As Integer, x As Double, dx As Double, fp As Double, i As Integer, fc As Double 

    nbb = 0 

    x = x1 

    dx = (x2 - x1) / n 

    fp = Application.Run(Func, x) 

    For i = 1 To n 

        x = x + dx 

        fc = Application.Run(Func, x) 

        If fc * fp <= 0 Then 

            nbb = nbb + 1 

            ReDim Preserve xb1(1 To nbb), xb2(1 To nbb) 

            xb1(nbb) = x - dx 

            xb2(nbb) = x 

            If nbb = nb Then 

                nb = nbb 

                Exit Sub 

            End If 

        End If 

        fp = fc 

    Next i 

    nb = nbb 

End Sub 

